

Using .FUNC Command




Table of Contents

1 Introduction 3

2 Demo Project..... 5

2.1 How Does .FUNC Command Work 5

2.2 How to Make Defined Functions Available for Use in PSpice..... 6

3 Bibliography 8

1 Introduction

It is known that you can use the keyword PARAMS: to define parameters to pass values into subcircuits as arguments and use them in expressions inside the subcircuit.

For example:


```
.subckt my_diode a k PARAMS:
+ p1=3
R_R1 a 1 1u
D_D1 1 k my_D
.MODEL my_D D
+ IS=3.3179E-6
+ N=3.5120
+ RS=1.0000E-3
+ CJO=80.00E-12
+ M=.31349
+ VJ=.3905
+ ISR=2.9061E-6
+ BV={2*p1}
+ IBV=.51729
+ TT=312.00E-9
.ENDS
```

Note


BV: Reverse Breakdown Knee Voltage

In the above subcircuit, p1 is defined as a parameter with the value 3 and used as an argument in the expression $2 * p1$. This defines that the reverse breakdown voltage of the diode my_D is 6V.

Assuming there is a simple test circuit as follows:



A transient simulation result could be displayed in the following figure.



So far it works fine. But sometimes expressions inside the subcircuit could be very complex and become quite long, e.g. $BV=\{\text{LOG}_{10}(2*p1+4)+p2+\text{EXP}(p3/5)\}$.

Note

LOG10(x): log base 10 in PSpice

p1, p2, p3: e.g. defined parameters in subcircuit

EXP(x): exponential function

For some reasons, you don't want to use the complex expression $\text{LOG}_{10}(2*p1+4)+p2+\text{EXP}(p3/5)$ to describe your model parameter BV, but prefer a short one. A typical consideration could be:

```
.subckt my_diode a k PARAMS:
+ p1=3
+ p2=5
+ p3=0
+ p4={LOG10(2*p1+4)+p2+EXP(p3/5)}
R_R1 a 1 1u
D_D1 1 k my_D
.MODEL my_D D
+ IS=3.3179E-6
+ N=3.5120
+ RS=1.0000E-3
+ CJO=80.00E-12
+ M=.31349
+ VJ=.3905
+ ISR=2.9061E-6
+ BV={p4}
+ IBV=.51729
+ TT=312.00E-9
.ENDS
```

If you try to run the simulation, you should get errors in PSpice Output file as follows:

- ERROR(ORPSIM-15167): Undefined parameter: p3.
- ERROR(ORPSIM-15167): Undefined parameter: p2.
- ERROR(ORPSIM-15167): Undefined parameter: p1.

How to solve this problem?

2 Demo Project

The nested parameters (e.g. p3, p2 and p1) cannot be recognised in PSpice. Based on the same test circuit we can change the definition of the subcircuit as follows:

```
.subckt my_diode a k PARAMS:  
+ p1=3  
+ p2=5  
+ p3=0  
R_R1 a 1 1u  
D_D1 1 k my_D  
.MODEL my_D D  
+ IS=3.3179E-6  
+ N=3.5120  
+ RS=1.0000E-3  
+ CJO=80.00E-12  
+ M=.31349  
+ VJ=.3905  
+ ISR=2.9061E-6  
+ BV={my_bv(p1,p2,p3)}  
+ IBV=.51729  
+ TT=312.00E-9  
.ENDS
```

Here we used a function `my_bv(p1,p2,p3)` to assign the value to the model parameter BV. In a texteditor we can define a function using the command `.FUNC` and save it as e.g. `function.inc`.

```
.FUNC my_bv(p1,p2,p3) {LOG10(2*p1+4)+p2+EXP(p3/5)}
```

Note

`.FUNC` commands can also be defined in subcircuits directly. In those cases, they only have local scope.

2.1 How Does .FUNC Command Work

Purpose

The `.FUNC` command defines functions used in expressions. Besides their obvious flexibility, they are useful for where there are several similar sub expressions in a circuit file.

General form

```
.FUNC <name>([arg]*) {<body>}
```

Examples

```
.FUNC E(x) {exp(x)}
```

```
.FUNC DECAY(CNST) {E(-CNST*TIME)}
```

```
.FUNC MIN3(A,B,C) {MIN(A,MIN(B,C))}
```

Arguments and options

- **.FUNC**
Functions cannot be redefined and the function name must not be the same as any of the predefined functions (e.g., SIN and SQRT). See **Numeric expression conventions** in PSpice Reference Guide for a list of valid expressions.
- **[arg]**
Specifies up to 10 arguments in a definition. The number of arguments in the use of a function must agree with the number in the definition. The arguments cannot be node names.
- **<body>**
The <body> of a defined function is handled in the same way as any math expression; it is enclosed in curly braces {}.


2.2 How to Make Defined Functions Available for Use in PSpice

We have already defined the following function:

```
.FUNC my_bv(p1,p2,p3) {LOG10(2*p1+4)+p2+EXP(p3/5)}
```

Where p1, p2 and p3 are the arguments of the function my_bv and LOG10(2*p1+4)+p2+EXP(p3/5) is the associated mathematic expression.

Open the **Simulation Settings** window and click the tap **Configuration Files > Include > Browse...** to find the saved function.inc file. Then you can configure this file globally or locally. In our example the function.inc file is configured locally to the current design.



Note

Include file: An include file is a user-defined file that contains:


- PSpice commands, or
- Supplemental text comments that you want to appear in the PSpice output file


Example

An include file that contains definitions, using the PSpice .FUNC command, for functions that you want to use in numeric expressions elsewhere in your design. You can create an include file using any text editor, such as Notepad. Typically, include file names have an .INC extension.

Run the simulation again. Now PSpice should work fine and you can observe the result.

If you set **All** for **Data Collection Options** as shown in the image, you can display the voltage for the diode within the subcircuit directly using e.g. V2(X_D1.D_D1).





3 Bibliography

- [1] PSpice User Guide, Cadence
- [2] PSpice Reference Guide, Cadence
- [3] OrCAD Capture User Guide, Cadence